

max65 User Guide

max65 is a command-line macro cross-assembler for the 65xx CPU family. It is useful for many systems but it specifically targets 8-bit Acorn computers like the Electron and BBC Micro.

This user guide explains how to use **max65** but is not a tutorial on 65xx assembly programming. There are many books and online resources about that.

The entire **max65** package is Copyright © 2022-2023 [0xC0DE](#) (0xC0DE6502@gmail.com). All rights reserved.

Table of contents

[Example program](#)

[Overview](#)

[Command-line options](#)

[Error and warning messages](#)

[Expressions](#)

[Literals](#)

[Symbols](#)

[Predefined global symbols](#)

[Symbol scopes](#)

[Operators](#)

[Built-in functions](#)

[Assembler directives](#)

[Macros](#)

[Comparing with BeebAsm](#)

[Features beyond BeebAsm](#)

[Quirks and tips](#)

[Download and install](#)

[Linux and macOS](#)

[Changelog](#)

[Disclaimer](#)

[Contact](#)

Example program

Let's dive right in with a short but real program to demonstrate some of the features of **max65**:

```
\ define some constants
RUN_ADDR=$0400          ; binary will execute at this address
LOAD_ADDR=$2000         ; binary will load at this address
OFFSET=LOAD_ADDR-RUN_ADDR ; displacement between main program and relocater

\ define zeropage variables
org 0                   ; start assembling at address $0000
.sin skip 2             ; reserve 2 bytes for zeropage variable 'sin'
```

```

\ main program
org RUN_ADDR          ; continue assembling at address 'RUN_ADDR'
.main                ; named label 'main': start of main program
  lda #<S: sta sin+0    ; write low byte of 'S' to zeropage variable 'sin'
  lda #>S: sta sin+1    ; write high byte of 'S' to zeropage variable 'sin'
  ; defining 'S' *after* using it is perfectly legal (lazy expression evaluation)
  S=(1<<16)*sin(rad(deg(45))) ; set 'S': use complex expressions and built-in
functions
  rts                  ; return to caller
.end                  ; end of main program

\ relocater stub that moves main program from 'LOAD_ADDR' to 'RUN_ADDR'
org *, *+OFFSET      ; continue assembling at current PC (*)
                    ; but also set logical PC to where relocater will
execute
.entry              ; program entry (execution starts here)
  ldx #end-main      ; number of bytes to copy (size of main program)
  .                  ; anonymous (unnamed) label
  lda LOAD_ADDR-1,x  ; copy byte from 'LOAD_ADDR' area
  sta RUN_ADDR-1,x   ;          to 'RUN_ADDR' area
  dex                ; point register X to next byte
  bne -              ; branch to previously defined anonymous label
  jmp main           ; finally, execute main program at 'RUN_ADDR'

; save the final binary named "CODE" from label 'main' to current PC (*)
; also save its accompanying .INF file
; set execution address to 'entry' (the $FF0000 signifies a host address)
; set load address to 'LOAD_ADDR' (again, the $FF0000 signifies a host address)
save "CODE", main, *, $FF0000|entry, $FF0000|LOAD_ADDR

```

Overview

max65 is heavily inspired by BeebAsm and aims to improve and extend it. If you are familiar with BeebAsm and/or BBC BASIC, many assembler directives and built-in functions are instantly recognisable. For a quick start, see [Comparing with BeebAsm](#) and [Features beyond BeebAsm](#).

max65 is a two-pass assembler. In pass 1 source files are tokenised and tokens are parsed to internal commands. In pass 2 these internal commands are translated to machine code and data. You can then save any part (or parts) of the 64Kb address space to your computer as raw binary file(s).

A source file is a plain text file consisting of 65xx instructions, assembler directives, label definitions and symbol assignments. Each element may be put on a separate line. A single line may also contain multiple elements, separated by colons (:). A source file also usually contains whitespace and comments.

This example shows all elements in a source file:

```

org &e00          ; assembler directive
.start           ; label definition
NUM=%1000_0000    ; symbol assignment
lda #NUM>>2      ; 65xx instruction

```

`max65` uses lazy expression evaluation. The main benefit for the user is that any symbol may be used (forward referenced) before being defined. The only exception is that macros must be defined before use. In pass 2 the final evaluation of expressions is done and all symbols must be resolved then of course. Example of forward referencing:

```
; All symbols below are used before being defined
lda data,x
sta &5800+N*320
rts
.data equb D1, D2, D3
D1=D3-N: D2=D1*2: D3=42: N=10
```

In some cases `max65` needs to make an educated guess in pass 1 about forward references, especially when the choice between zeropage and absolute addressing modes needs to be made. It is therefore recommended (but not required) to define zeropage labels as early as possible in your program.

Command-line options

Here is a brief summary of how to invoke `max65`:

```
max65 [-h] [-l <listfile>] <infile>
```

Option	Description
<infile>	Input file (plain text source)
-l <listfile>	Create a listing file
-h	Show a help message and exit

Example command-line: `max65 main.6502 -l main-listing.txt`.

Error and warning messages

When all goes well `max65` happily assembles the source file and exits with exit code 0 (success). However, when `max65` fails to assemble the source file, an error message is written to the standard error file (stderr) and the assembler exits with exit code 1 (failure). An error message shows the source filename, line number and cause of the error. Example of a typical error message: `*** error in pass 2: file main.6502, line 10: operator '/' doesn't work on strings`.

In a few cases `max65` writes a warning message to stderr, but continues assembling your program. Example of a typical warning message: `*** warning in pass 2: file main.6502, line 6: instruction 'lda' can be assembled 1 byte shorter in zeropage addressing mode`.

Expressions

max65 can handle arbitrarily complex expressions in directives, symbol assignments or 65xx instructions. Expressions consist of literals, symbols, operators and built-in functions. Each expression must eventually evaluate to an integer, float or string in pass 2.

Literals

You can use integer, float and string literals in expressions.

- **Integers** are positive or negative whole numbers, of just about any size. In practice you will use 8-bit, 16-bit or 32-bit integers though. You may use decimal notation (e.g. `123`, `-64`), hexadecimal notation (e.g. `$FFEE`, `-&ac`), binary notation (e.g. `%1101`, `-%11110`) or char notation (e.g. `'a'`, `-'C'`). Use an underscore (`_`) or a single quote (`'`) to group digits, e.g. `%110_001`, `&FE'03`, `12'34_56`. Some chars need to be escaped with a backslash (`\`): `'\\'`, `'\'`. Empty chars (`' '`) are not allowed.
- **Floats** are positive or negative fractional numbers, of just about any size. Only decimal notation is supported with at least 1 digit before and after the decimal point, e.g. `3.1415`, `-5.0`, but not `.001`. Scientific notation (e.g. `1.23E-6`) is not supported. In many cases only the integer part of a float will be used automatically. For instance, `ldx #3.14` will be assembled as `ldx #3`.
- **Strings** are sequences of 0 or more characters, enclosed in double quotation marks. Examples: `""`, `"Hello!"`, `"This is a backslash: \\"`. Unlike an empty char (`' '`), an empty string (`""`) is perfectly allowed. Some characters in a string need to be escaped with a backslash (`\`): `'\\'`, `'\''`.

Symbols

Symbols are used to name things in a source file. The name of a symbol is case sensitive and consists of any mix of digits, letters and underscores. It must not start with a digit though. Examples of valid symbol names are: `_`, `Lives`, `data_Table4`. A symbol has a [specific scope](#) in which it is valid and it (eventually) always refers to an integer, float or string. There are some [predefined global symbols](#).

Once defined, the value of a symbol cannot be changed by the programmer.

A symbol is defined in 1 of 5 ways:

1. **Label definition.** A label is defined by a period (`.`) directly followed by a symbol name, e.g. `.sine256`, `.current_level`. The symbol is added to the internal nametable at the current scope level, unless it already exists there. Its value is set to the current program counter (PC) which is a 16-bit integer. You can also define anonymous (unnamed) labels by simply using a period without name:

```
.          ; anonymous label #1
tya
beq +      ; jump to anonymous label #2
txa
beq ++     ; jump to anonymous label #3
.          ; anonymous label #2
iny
bpl --     ; jump to anonymous label #1
.          ; anonymous label #3
```

```
dex
bmi --- ; jump to anonymous label #1
```

2. **Symbol assignment.** An assignment consists of a symbol name, followed by an equal sign (=), and an expression. Examples: `N=3`, `start_addr=&5800+N*&140`. The expression stored with the symbol in the nametable can contain forward references but must eventually evaluate to an integer, float or string.
3. **Assembler directive `for`.** A `for...next` loop defines a local scope for every cycle of the loop with the `for`-loop variable (integer or float) as a local symbol. Example: `for n, 1, 10 ... next`. The body of the loop is assembled 10 times within the context of that local scope. The local symbol `n` increments from 1 to 10 during that time.
4. **Macro definition.** When you define a macro, e.g. `macro add num1, num2 ... endmacro`, a special global symbol based on the macro name is created in the internal nametable, unless a macro with that exact name already exists. Macro names never clash with other symbols and also don't evaluate to any value.
5. **Macro call.** When you invoke a previously defined macro, e.g. `add 12, 34`, a local scope is created with local symbols that are named after the parameters (if any) in the macro definition: `num1` and `num2` in this example. Their values are set to the macro call arguments, `12` and `34` in this example. The macro body is then assembled within the context of that local scope.

Predefined global symbols

`max65` has some predefined global symbols:

Symbol	Type	Value
<code>PI</code>	Float	3.141592653589793
<code>FALSE</code>	Integer	0
<code>TRUE</code>	Integer	-1
<code>*</code> (or <code>P%</code>)	Integer	Current PC
<code>@</code>	Integer	Current logical PC
<code>VERSION</code>	Integer	Version of <code>max65</code> , e.g. \$010A is version 1.10

Symbol scopes

You can create virtually unlimited nested local scopes for your symbols using curly braces, e.g.:

```
N=-1                ; global symbol 'N'
.lab                ; global label 'lab'
{
  N=3                ; local symbol 'N', overrides global symbol 'N'
  for n, 1, N: equib n: next ; local symbol 'n' for each loop cycle
  { .lab print N: N=6 } ; local label 'lab' and another local symbol 'N'
                        ; overrides global label 'lab'
```

```
; and overrides symbol 'N' in parent scope
```

```
}
```

Defining symbols for a different scope within the current scope is also possible by using a special notation, e.g.:

```
{
  N*=32      ; define global symbol 'N'
  {
    .^lab     ; define local label 'lab' that is visible in current and parent
scope
    .*lab2    ; define global label 'lab2'
    { N^=-99 } ; define local symbol 'N' that is visible in current and parent
scope
    print N   ; -99
  }
}
print N      ; 32
```

Operators

Operators in order of increasing precedence:

Operator	Associativity	Description
<, >	Right	Equal to <code>lo()</code> and <code>hi()</code> respectively, e.g. <code>lda #<&5800+n*320</code> is the same as <code>lda #lo(&5800+n*320)</code>
or	Left	Logical OR, e.g. <code>if addr>=&5800 or addr<&3000 ... endif</code>
and	Left	Logical AND, e.g. <code>if N==3 and DEBUG ... endif</code>
\	Left	Bitwise OR, e.g. <code>lda #1\ 2\ 4\ &80</code>
^	Left	Bitwise XOR (EOR), e.g. <code>ldx #n^255</code>
&	Left	Bitwise AND, e.g. <code>lda #N&3</code>
==, =, !=, <>	Left	(In)equality, e.g. <code>assert N!=3 and DBG=2</code>
<, <=, >, >=	Left	Comparison, e.g. <code>if addr>=&5800 or addr<&3000 ... endif</code>
+, -	Left	Addition and subtraction, e.g. <code>equw 4*WIDTH-(N+8)</code> , <code>print "S="+lower\$(S)</code>
, /, div, mod	Left	Multiplication and division, e.g. <code>equb 8(y div 8)</code> , <code>S=3*"ABC"</code>
<<, >>	Left	Bit and string shift, e.g. <code>lda #1<<5-1, equs "Hello world">>6</code>

Operator	Associativity	Description
<code>not, ~, -</code>	Right	Logical NOT, bitwise NOT and unary minus, e.g. <code>if not defined("N") ... endif, lda #Q&~3</code>

Built-in functions

Built-in functions and expressions in parentheses (or, alternatively, in square brackets) have the highest priority:

Function	Description
<code>lo()</code>	Least significant byte (lower 8 bits), e.g. <code>adc #lo(SCRN)</code>
<code>hi()</code>	Most significant byte (upper 8 bits). Technically bits 15..8. E.g. <code>equb hi(SCRN+3*&140)</code>
<code>sin()</code>	Sine of an angle in radians, e.g. <code>equw 512*sin(t)</code>
<code>cos()</code>	Cosine of an angle in radians, e.g. <code>N=1024*cos(PI/4)</code>
<code>tan()</code>	Tangent of an angle in radians, e.g. <code>equd 32*tan(2*t)</code>
<code>asn()</code>	Arc sine (only defined for domain [-1, 1]), e.g. <code>n=asn(-0.5)</code>
<code>acs()</code>	Arc cosine (only defined for domain [-1, 1]), e.g. <code>n=acs(0.3*x)</code>
<code>atn()</code>	Arc tangent, e.g. <code>T=atn(100)</code>
<code>deg()</code>	Convert radians to degrees, e.g. <code>d=deg(PI/8)</code>
<code>rad()</code>	Convert degrees to radians, e.g. <code>r=rad(180)</code>
<code>int()</code>	Truncate float to integer, e.g. <code>lda #int(3.14)</code>
<code>abs()</code>	Absolute value, e.g. <code>m=abs(sin(t))</code>
<code>sqr()</code>	Square root (only defined for values ≥ 0), e.g. <code>rt=sqr(36)</code>
<code>sgn()</code>	Sign of its argument: -1 for negative values, 0 for zero, 1 for positive values. E.g. <code>p=x+16*sgn(n)</code>
<code>log()</code>	Logarithm (base 10, only defined for values > 0), e.g. <code>equb log(1000)</code>
<code>ln()</code>	Natural logarithm (base e, only defined for values > 0), e.g. <code>equb ln(x/3)</code>
<code>exp()</code>	e raised to the power of the argument, e.g. <code>e=exp(1)</code>
<code>rnd()</code>	Random number. Only defined for integer values > 0 . <code>rnd(1)</code> returns a float in the range [0, 1]. <code>rnd(n)</code> with integer $n > 1$ returns an integer in the range [1, n]. You can seed the random number generator with the <code>randomize</code> directive
<code>val()</code>	String to number. Checks if a string starts with a valid integer or float and returns that (or 0 if no number was found). E.g. <code>P=val("-3.14PI!")</code>
<code>len()</code>	Length of a string, e.g. <code>print len("Hello!")</code>
<code>asc()</code>	ASCII value of first character of a string (or -1 for an empty string), e.g. <code>val=asc("Hello!")</code>

Function	Description
<code>str\$()</code>	Convert number to a string, e.g. <code>S=str\$(123.4)>>1</code>
<code>str\$~()</code>	Convert number to a string with the number in hexadecimal format, e.g. <code>print "\$", str\$~(128)</code> . Note: negative numbers are not printed in two's complement, so for example <code>str\$~(-140)</code> translates to the string <code>"-8C"</code> , not <code>"FFFFFF74"</code> (which assumes 32-bit)
<code>chr\$()</code>	Convert an ASCII value to a string of length 1 containing that ASCII character. Only valid for domain [32, 126]. E.g. <code>S=chr\$(122)</code>
<code>lower\$()</code>	Convert a string to lowercase, e.g. <code>S=lower\$("ABC123")</code>
<code>upper\$()</code>	Convert a string to uppercase, e.g. <code>S=upper\$("abc123")</code>
<code>time\$()</code>	Date and time of assembly (string). The argument is a string that determines the date/time format as specified by the Python (or C) function <code>strftime()</code> . <code>time\$("")</code> returns date/time formatted as <code>"%a,%d %b %Y.%H:%M:%S"</code> , e.g. <code>"Mon,16 Jan 2023.17:46:03"</code>
<code>defined()</code>	Check if symbol is defined (returns <code>TRUE</code>) or not (returns <code>FALSE</code>), e.g. <code>if not defined("SCRN") ... endif</code>

Assembler directives

Directives (or pseudo-ops) control the assembly process. A directive takes zero or more arguments. For instance, `skip 64` tells `max65` to skip 64 bytes. An argument is an expression that must evaluate to an integer, float or string. Floats are truncated when an integer is expected.

Sometimes the value of an argument must be known in pass 1 to let the assembler make the right decision. This means expressions must not contain unresolved symbols (forward references). For instance, in an `include` directive, `max65` must immediately know the name of the source file to include.

Other directives, like `print`, only evaluate their arguments in pass 2. In this case expressions that still have unresolved symbols (forward references) after pass 1 are allowed.

Directives that evaluate their arguments in pass 1:

Directive	Description
<code>align expr</code>	Increment PC to next multiple of <code>expr</code> , e.g. <code>align 256</code> aligns PC to the next memory page
<code>elif expr</code>	Mark the start of an <code>elif</code> -block in an <code>if...endif</code> . See <code>if</code>
<code>else</code>	Mark the start of an <code>else</code> -block in an <code>if...endif</code> . See <code>if</code>
<code>endif</code>	Mark the end of an <code>if...endif</code> block. See <code>if</code>
<code>endmacro</code>	Mark the end of a macro definition. See <code>macro</code>
<code>equb expr_1 [, expr_2, ..., expr_n]</code>	Insert one or more bytes and/or strings, e.g. <code>equb "Hello!", 13, 10, 0</code> . For numeric arguments forward references are allowed

Directive	Description
equs	Equivalent to equb
for <i>sym</i> , <i>expr_1</i> , <i>expr_2</i> [, <i>expr_3</i>] ... next	Assemble a block of code/data one or more times. The loop counter <i>sym</i> is a local symbol which changes from <i>expr_1</i> to <i>expr_2</i> (inclusive) in steps of <i>expr_3</i> (-1 or 1 if unspecified). Examples: for <i>n</i> , 0, 9: equb <i>n</i> : next . Or: for <i>f</i> , 3.5, -1.5, -0.75: print <i>f</i> : next
if <i>expr_1</i> ... [elif <i>expr_2</i> ...] ... [elif <i>expr_n</i> ...] [else ...] endif	Conditional assembly. Assemble the code/data block for which the corresponding if -condition or the first elif -condition (in order of appearance) evaluates to a non-zero number (TRUE). When none exists, assemble the else -block (if any). Examples: if <i>n</i> >3: print <i>n</i> : endif . Or: if <i>n</i> >3: print <i>n</i> : elif <i>n</i> <-3: print - <i>n</i> : else : print "Invalid": endif
incbin <i>expr</i>	Insert binary file named <i>expr</i> , e.g. incbin "data/table.bin". <i>expr</i> is a string that contains a valid path to an existing file
include <i>expr</i>	Assemble and insert source file named <i>expr</i> , e.g. include "src/spriteplot.asm". <i>expr</i> is a string that contains a valid path to an existing file
macro <i>sym_1</i> [, <i>sym_2</i> , ..., <i>sym_n</i>] ... endmacro	Define a macro named <i>sym_1</i> with optional parameters named <i>sym_2</i> , ..., <i>sym_n</i> . See also: Macros
next	Mark the end of a for...next loop. See for
org <i>expr_1</i> [, <i>expr_2</i>]	Set PC (where code/data is assembled) to <i>expr_1</i> . Also set the logical PC (on which labels are based) to <i>expr_2</i> . If <i>expr_2</i> is not specified, the logical PC will be equal to PC. Examples: org \$1000. Or: org \$1000, \$2000
skip <i>expr</i>	Increment PC by <i>expr</i> bytes, e.g. skip 16

Directives that evaluate their arguments in pass 2:

Directive	Description
assert <i>expr_1</i> [, <i>expr_2</i> , ..., <i>expr_n</i>]	Evaluate one or more arguments and trigger an error for the first expression (in order of appearance) that is zero (FALSE). Example: assert <i>N</i> <128, <i>P</i> %<\$1000
error [<i>expr_1</i> , ..., <i>expr_n</i>]	Trigger an error after printing <i>expr_1</i> , ..., <i>expr_n</i> to the standard error file (stderr). Example: if <i>N</i> >=128: error "Expected <i>N</i> <128, but <i>N</i> =", <i>N</i> : endif

Directive	Description
equd <i>expr_1</i> [, <i>expr_2</i> , ..., <i>expr_n</i>]	Insert one or more double words (32-bit integers), e.g. <code>equd \$C0DE6502</code>
equw <i>expr_1</i> [, <i>expr_2</i> , ..., <i>expr_n</i>]	Insert one or more words (16-bit integers), e.g. <code>equw \$C0DE</code>
guard [<i>expr_1</i> , ..., <i>expr_n</i>]	Set multiple guards at addresses <i>expr_1</i> , ..., <i>expr_n</i> . An address guard triggers an error when code/data is assembled at that address. When no arguments are given, clear all current guards. Example: <code>guard \$5800, \$8000</code>
print [<i>expr_1</i> , ..., <i>expr_n</i>]	Print zero or more expressions <i>expr_1</i> , ..., <i>expr_n</i> to the standard output file (stdout) and finish with a newline. When no arguments are given, just print a newline
randomize <i>expr</i>	Seed the random number generator with <i>expr</i> , e.g. <code>randomize 12345</code> . <i>expr</i> can be any integer, float or string
save <i>expr_1</i> , <i>expr_2</i> , <i>expr_3</i> [, <i>expr_4</i> [, <i>expr_5</i>]]	Save a code/data block from the 64Kb address space to a raw binary file named <i>expr_1</i> . The block starts at <i>expr_2</i> and ends at <i>expr_3</i> (exclusive). The optional execution address <i>expr_4</i> and load address <i>expr_5</i> are used in the accompanying .inf file. When not specified, these are equal to <i>expr_2</i> . Example: <code>save "CODE", \$1000, \$2000, \$f25, \$e00</code>

Macros

Macros are user defined code/data blocks and can be inserted anywhere in your program using a macro call. A macro is always global and must be defined before use. It takes zero or more parameters. Here is an example of a macro definition and a macro call:

```
\ macro definition
macro add_ptr ptr, val    ; 2 parameters
    lda ptr
    clc
    adc #lo(val)
    sta ptr
    lda ptr+1
    adc #hi(val)
    sta ptr+1
endmacro
```

```
\ macro call
add_ptr &70, &140          ; 2 arguments are passed to the macro
```

The macro call defines a local scope and binds the given arguments to the macro parameters. The macro call above therefore expands to:

```
{
  ptr=&70
  val=&140
  lda ptr
  clc
  adc #lo(val)
  sta ptr
  lda ptr+1
  adc #hi(val)
  sta ptr+1
}
```

A macro can call other macros and even itself recursively.

Comparing with BeebAsm

max65 follows BeebAsm's syntax closely, but there are some differences and alternatives:

BeebAsm	max65
LEFT\$("abcde", 3) (equals "abc")	"abcde">>2
RIGHT\$("abcde", 3) (equals "cde")	"abcde"<<2
MID\$("abcde", 2, 3) (equals "bcd")	"abcde"<<1>>1
STRING\$(3, "ABC") (equals "ABCABCABC")	"ABC"*3
N=?3 (define N if not defined yet)	if not defined("N"): N=3: endif
COPYBLOCK \$1000, \$1100, \$500	org \$500, \$1000
x^y (raise to the power of)	exp(y*ln(x)) for x>0
EVAL()	N/A
TIME\$	TIME\$("")
CPU	N/A
AND (logical)	and
AND (bitwise)	&
OR (logical)	or
OR (bitwise)	\

BeebAsm	max65
EOR (logical)	N/A
EOR (bitwise)	<code>^</code>
NOT (logical)	<code>not</code>
NOT (bitwise)	<code>~</code>
<code>SKIPTO \$2000</code>	<code>org \$2000</code>
<code>CLEAR</code>	<code>guard</code> (without args) clears all guards only
<code>MAPCHAR</code>	N/A
<code>PRINT ~200</code> (equals <code>"&C8"</code>)	<code>print "&"+str\$~(200)</code>
<code>ASM()</code>	N/A
<code>FILELINE\$</code>	N/A
<code>CALLSTACK\$</code>	N/A
<code>"AB""CD"</code> (quote doubling)	<code>"AB\"CD"</code> (escape char)
<code>PUTTEXT</code>	N/A (no .ssd disk image I/O)
<code>PUTFILE</code>	N/A (no .ssd disk image I/O)
<code>PUTBASIC</code>	N/A (no .ssd disk image I/O)
<code>RND(100)</code> (random integer x, 0<=x<=99)	1<=x<=100 (like BBC BASIC)

Features beyond BeebAsm

max65 extends or improves BeebAsm in several ways:

Feature	Description
Undocumented 6502 instructions	<code>alr, anc, ane, arr, dcp, dop, hlt, isb, las, lax, nop, rla, rra, sax, sbc, sbx, sha, shx, shy, slo, sre, tas, top</code>
<code>N^=3</code>	Define N in parent scope
<code>N*=3</code>	Define N in global scope
<code>&C0_DE,</code> <code>\$6'502, 1'234,</code> <code>3.14_15,</code> <code>%00_10'00</code>	Digit grouping with <code>_</code> and <code>'</code> for all numbers, not just binary
<code>N=A*A: A=2</code> (forward references in assignments)	Lazy expression evaluation allows forward references everywhere

Feature	Description
<code>N=A*A: A=2*N</code>	Error on detection of circular references
<code>macro m: m: endmacro: m</code>	Error on detection of runaway recursion
in file a.asm: <code>include "a.asm"</code>	Error on detection of circular <code>includes</code>
<code>org \$200, \$500</code>	Optional second argument in <code>org</code> directive sets logical PC (@), e.g. assemble at \$200, but labels are based on \$500. Almost like <code>COPYBLOCK</code> in BeebAsm, or <code>0%</code> in BBC BASIC
<code><<, >></code> and <code>*</code> work on strings	<code>"abc"<<1</code> equals <code>"bc"</code> , <code>"abc">>2</code> equals <code>"a"</code> , <code>"A"*3</code> equals <code>"AAA"</code>
<code>{ ... include ... }</code>	Including source files works on any scope level (curly braces), and inside <code>for...next</code> loops as well
<code>randomize</code>	Seed the random number generator with any integer, float or string
<code>error</code>	Accepts zero or more arguments so it works similar to the <code>print</code> directive
<code>defined("N")</code>	<code>TRUE</code> if symbol <code>N</code> is defined, <code>FALSE</code> otherwise
<code>guard</code>	The <code>guard</code> directive sets one or more guards on the supplied memory addresses. When no arguments are given, all guards are cleared
zeropage vs absolute	<code>max65</code> issues a friendly warning when an instruction could have used zeropage addressing mode (saving 1 byte)
<code>.</code> (anonymous labels)	Use <code>.</code> to define unnamed labels. Relative branch instructions can jump backward or forward to them, e.g. <code>bne -</code> , or <code>bpl ++</code>
numbers	When an integer is expected, a float number is automatically truncated (not rounded) to an integer. Large integers and negative integers are allowed for 65xx instructions and directives like <code>equb/equw/equd</code> . E.g. <code>lda #-2</code> is equal to <code>lda #&fe</code> , <code>ldx #&123</code> is equal to <code>ldx #&23</code> (lower 8 bits), <code>equw \$123456</code> is equal to <code>equw \$3456</code> (lower 16 bits).

Quirks and tips

- You can use `/` and `\\` in file paths, e.g. `include "src/prog.asm"`, `incbin "..\\data.bin"`.
- In an `if`-block or `elif`-block where the condition evaluates to zero (`FALSE`), chars and strings still need to be valid because of how the tokeniser works.
- Everything is case insensitive except for symbols which are case sensitive.

Download and install

The latest release of `max65` can always be found on [GitHub](#).

Download the .zip file, extract it and optionally add `max65.exe` to your system path. The assembler is now ready for use.

The assembler is compiled and tested on 64-bit Windows 11. It is fully portable as long as you keep `max65.exe` together with the accompanying `python*.dll` and `python*.zip` files.

Linux and macOS

`max65` is known to work on Ubuntu 20.04.5 with Wine 5.0-3 and on Ubuntu 22.04.1 with Wine 6.0.3. I am confident that other combinations of Linux and Wine will work equally well. This particular construction is untested on macOS.

Changelog

Version	Date	Changes
0.11	Feb 17, 2023	Create optional listing file (-l)
0.10	Feb 15, 2023	Initial release

Disclaimer

The author, 0xC0DE, of this software accepts no responsibility for damages resulting from the use of this product and makes no warranty or representation, either express or implied, including but not limited to, any implied warranty of merchantability or fitness for a particular purpose. This software is provided "AS IS", and you, its user, assume all risks when using it.

Contact

If you have any questions, suggestions or bug reports about `max65`, please contact me at 0xC0DE6502@gmail.com or on Twitter [@0xC0DE6502](https://twitter.com/0xC0DE6502).